<commit_before><?xml version="1.0" encoding="UTF-8"?>
<xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
                xmlns:tei="http://www.tei-c.org/ns/1.0"
                xmlns:rails="http://svario.it/rails-xslt"
                exclude-result-prefixes="tei rails"
                version="1.0">
	<xsl:variable name="side-data" select="//rails:side-data"/>

	<xsl:variable name="space-char" xml:space="preserve"><xsl:text>&#32;</xsl:text></xsl:variable>

	<xsl:template match="/*">
		<xsl:element name="dl">
			<!-- FIXME: find a better way to show lemmas and transliterations -->
			<xsl:apply-templates select="$side-data/rails:lemma"/>
			<xsl:element name="div">
				<xsl:attribute name="class">transliterations</xsl:attribute>

				<xsl:text>(</xsl:text>
				<xsl:apply-templates select="$side-data/rails:transliterations/rails:*"/>
				<xsl:text>)</xsl:text>
			</xsl:element>

			<xsl:apply-templates select="tei:sense"/>
		</xsl:element>
	</xsl:template>

	<xsl:template match="rails:lemma">
		<xsl:element name="dt">
			<xsl:attribute name="class">lemma</xsl:attribute>
			<xsl:attribute name="xml:lang">san-Deva</xsl:attribute>

			<xsl:element name="a">
				<xsl:attribute name="href">
					<xsl:text>/lemma/</xsl:text>
					<xsl:value-of select="$side-data/rails:lemma"/>
				</xsl:attribute>

				<xsl:element name="dfn">
					<xsl:value-of select="$side-data/rails:lemma"/>
				</xsl:element>
			</xsl:element>
		</xsl:element>
	</xsl:template>

	<xsl:template match="rails:transliterations/rails:*">
		<xsl:variable name="method" select="local-name()"/>
		<xsl:variable name="lemma" select="text()"/>

		<xsl:element name="dt">
			<xsl:attribute name="class">transliteration</xsl:attribute>
			<xsl:attribute name="xml:lang">san-Latn</xsl:attribute> <!-- FIXME: set according to $method -->

			<xsl:element name="a">
				<xsl:attribute name="href">
					<xsl:text>/lemma/</xsl:text>
					<xsl:value-of select="$lemma"/>
					<xsl:text>?iscript=</xsl:text>
					<xsl:value-of select="$method"/>
				</xsl:attribute>

				<xsl:element name="span">
					<xsl:attribute name="class">method</xsl:attribute>
					<xsl:value-of select="$method"/>
				</xsl:element>
				<xsl:text>: </xsl:text>
				<xsl:value-of select="$lemma"/>
			</xsl:element>
		</xsl:element>
	</xsl:template>

	<xsl:template match="tei:sense">
		<dd class="sense">
			<xsl:apply-templates select="node()[not(self::tei:note)]"/>

			<xsl:call-template name="provenance-info">
				<xsl:with-param name="sense" select="."/>
			</xsl:call-template>
		</dd>
	</xsl:template>

	<xsl:template match="tei:gram">
		<em class="gram">
			<xsl:value-of select="."/>
		</em>
	</xsl:template>

	<xsl:template match="tei:cit">
		<!-- FIXME: use a proper citational URI scheme -->
		<a href="/biblio/FIXME">
			<xsl:value-of select="normalize-space(.)"/>
		</a>

		<xsl:value-of select="$space-char"/>
	</xsl:template>

	<xsl:template name="provenance-info">
		<xsl:param name="sense"/>
		<xsl:variable name="dictionary" select="$side-data/rails:dict"/>

		<xsl:value-of select="$space-char"/>

		<cite class="provenance-info">
			<xsl:text>(</xsl:text>
			<abbr class="dictionary">
				<a href="/dictionary/{$dictionary}">
					<xsl:value-of select="$dictionary"/>
				</a>
			</abbr>

			<xsl:text> at page </xsl:text>
			<!-- FIXME: link to the scan using a proper URI scheme -->
			<a href="/scan/FIXME">
				<xsl:value-of select="$sense/tei:note/tei:ref"/>
			</a>
			<xsl:text>)</xsl:text>
		</cite>
	</xsl:template>
</xsl:stylesheet><commit_msg>Link references to scans<commit_after><?xml version="1.0" encoding="UTF-8"?>
<xsl:stylesheet xmlns:xsl="http://www.w3.org/1999/XSL/Transform"
                xmlns:tei="http://www.tei-c.org/ns/1.0"
                xmlns:rails="http://svario.it/rails-xslt"
                exclude-result-prefixes="tei rails"
                version="1.0">
	<xsl:variable name="side-data" select="//rails:side-data"/>

	<xsl:variable name="space-char" xml:space="preserve"><xsl:text>&#32;</xsl:text></xsl:variable>

	<xsl:template match="/*">
		<xsl:element name="dl">
			<!-- FIXME: find a better way to show lemmas and transliterations -->
			<xsl:apply-templates select="$side-data/rails:lemma"/>
			<xsl:element name="div">
				<xsl:attribute name="class">transliterations</xsl:attribute>

				<xsl:text>(</xsl:text>
				<xsl:apply-templates select="$side-data/rails:transliterations/rails:*"/>
				<xsl:text>)</xsl:text>
			</xsl:element>

			<xsl:apply-templates select="tei:sense"/>
		</xsl:element>
	</xsl:template>

	<xsl:template match="rails:lemma">
		<xsl:element name="dt">
			<xsl:attribute name="class">lemma</xsl:attribute>
			<xsl:attribute name="xml:lang">san-Deva</xsl:attribute>

			<xsl:element name="a">
				<xsl:attribute name="href">
					<xsl:text>/lemma/</xsl:text>
					<xsl:value-of select="$side-data/rails:lemma"/>
				</xsl:attribute>

				<xsl:element name="dfn">
					<xsl:value-of select="$side-data/rails:lemma"/>
				</xsl:element>
			</xsl:element>
		</xsl:element>
	</xsl:template>

	<xsl:template match="rails:transliterations/rails:*">
		<xsl:variable name="method" select="local-name()"/>
		<xsl:variable name="lemma" select="text()"/>

		<xsl:element name="dt">
			<xsl:attribute name="class">transliteration</xsl:attribute>
			<xsl:attribute name="xml:lang">san-Latn</xsl:attribute> <!-- FIXME: set according to $method -->

			<xsl:element name="a">
				<xsl:attribute name="href">
					<xsl:text>/lemma/</xsl:text>
					<xsl:value-of select="$lemma"/>
					<xsl:text>?iscript=</xsl:text>
					<xsl:value-of select="$method"/>
				</xsl:attribute>

				<xsl:element name="span">
					<xsl:attribute name="class">method</xsl:attribute>
					<xsl:value-of select="$method"/>
				</xsl:element>
				<xsl:text>: </xsl:text>
				<xsl:value-of select="$lemma"/>
			</xsl:element>
		</xsl:element>
	</xsl:template>

	<xsl:template match="tei:sense">
		<dd class="sense">
			<xsl:apply-templates select="node()[not(self::tei:note)]"/>

			<xsl:call-template name="provenance-info">
				<xsl:with-param name="sense" select="."/>
			</xsl:call-template>
		</dd>
	</xsl:template>

	<xsl:template match="tei:gram">
		<em class="gram">
			<xsl:value-of select="."/>
		</em>
	</xsl:template>

	<xsl:template match="tei:cit">
		<!-- FIXME: use a proper citational URI scheme -->
		<a href="/biblio/FIXME">
			<xsl:value-of select="normalize-space(.)"/>
		</a>

		<xsl:value-of select="$space-char"/>
	</xsl:template>

	<xsl:template name="provenance-info">
		<xsl:param name="sense"/>

		<xsl:variable name="dictionary" select="$side-data/rails:dict"/>
		<xsl:variable name="page-info" select="$sense/tei:note/tei:ref"/>
		<xsl:variable name="page-num" select="substring-after($page-info/@target, '#')"/>

		<xsl:value-of select="$space-char"/>

		<cite class="provenance-info">
			<xsl:text>(</xsl:text>
			<abbr class="dictionary">
				<a href="/dictionary/{$dictionary}">
					<xsl:value-of select="$dictionary"/>
				</a>
			</abbr>

			<xsl:text> at page </xsl:text>
			<!-- FIXME: link to the scan using a proper URI scheme -->
			<a href="/dictionary/{$dictionary}/scan/{$page-num}">
				<xsl:value-of select="$page-info"/>
			</a>
			<xsl:text>)</xsl:text>
		</cite>
	</xsl:template>
</xsl:stylesheet>
{'fraction_non_alphanumeric': 0.21851749793331496, 'fraction_numerical': 0.0038578120694406173, 'mean_word_length': 4.009661835748792, 'pattern_counts': {'":': 0, '<': 231, '<?xml version=': 2, '>': 231, 'https://': 0, 'lorem ipsum': 0, 'www.': 4, 'xml': 14}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '4128008', 'n_tokens_mistral': 3165, 'n_tokens_neox': 2527, 'n_words': 398}